United States Department of the Interior

Geological Survey

FORTRAN Program to Compute Chemical Geothermometers for Geothermal Fluids

By

Amy Rapport

U. S. Geological Survey

Menlo Park, California

U. S. Geological Survey

Open-File Report 82-308

This report is preliminary and has not been reviewed for conformity with U. S. Geological Survey editorial standards. Any use of trade names is for descriptive purposes only and does not imply endorsement by the USGS

January, 1982

Preface

The GEOTHERM Information System is a comprehensive system of databases and software to maintain data on the geology, geochemistry and hydrology of geothermal systems, both nationally and internationally. The information system is made up of four databases and software to assist in data entry, quality assurance, operations, and to generate useful outputs for system users. The system is housed on the U.S. Geological Survey Amdahl computer and inquiries for retrievals from system databases may be made of the GEOTHERM staff at the address given below. The database on geochemistry is also available to subscribers of the General Electric Mark III System, an internationally available timeshare network. General Electric customer support is available at 800-638-8730. This report addresses a GEOTHERM user-support software package used to generate geothermometer values from data in the Sample File which contains geochemical information on fluids from geothermal springs and wells. Both the FORTRAN code and associated documentation have been included. The program has been carefully checked and used for some time but users are advised to spot check the results from it as computer software can be less fool proof than one might hope. Readers of this report are encouraged to inform the Data Base Manager, GEOTHERM Information System, U.S. Geological Survey, 345 Middlefield Rd. MS 84, Menlo Park, CA 94024, of any omissions or errors. Comments or requests for further information are invited.

James D. Bliss
Data Base Manager
GEOTHERM Information System

INTRODUCTION

Chemical geothermometers represent one method of estimating temperature conditions at depth in geothermal systems. However, geothermometer results are not valid unless reaction rates are sufficient to bring fluid and minerals in the confining rock to equilibruim at the temperature of the geothermal reservoir. At the same time, the water-mineral reactions must be sluggish enough so as not to re-equilibrate as the fluids move to the surface or to the point at which the sample was collected. Concentrations of elements used for geothermometers should be high enough for equilibruim processes to be valid (Ellis and Mahon, 1977).

Geothermometer results should always be evaluated with respect to geology, hydrology and chemical properties of the fluids. Variation between the geothermometers for the same sample may reflect different chemical species equilibrating at different points in the geothermal system (Fournier, 1981). Mixing of geothermal fluids with other subsurface water can radically modify the chemistry. Geothermometer results should therefore be used with care, and the limitations and strengths of each should be fully understood.

The FORTRAN program given here calculates geothermometer values using silica, sodium, potassium and calcium from the results of chemical analyses of geothermal waters. In the case of the Na-K-Ca geothermometer, a correction is available using magnesium. This operation is described in Fournier and Potter (1978).

This program was specifically designed to be used with data from the GEOTHERM Sample file. The portions of the program that deal with the peculiarities of the GEOTHERM format are explained and a modified version of the program, suitable for use with more generalized data, is given. This paper documents the program so that it can be used by those with a rudimentary understanding of FORTRAN.

BACKGROUND OF PROGRAM

The geothermometer program was developed in order to facilitate the U. S. Geological Survey's assessment of low-temperature geothermal resources of the United States. The source of input data for the program is the Sample File. This program requires input data and generates some output which may or may not be necessary for general applications. In addition to the chemical constituents used to calculate the geothermometers, the input and output include the GEOTHERM record number; the temperature of the sample; the name of the sample source; the state, county, township, and range in which the source is located; the depth if the source is a well; pH; total dissolved solids; and the flow rate. The program calculates the total dissolved solids, the silica and the cation geothermometers and calculates and applies a magnesium correction when appropriate.

The equations used in the program are taken from Fournier (1977) and are given below:

GEOTHERMOMETERS

Silica

Quartz (conductive) $$TQC = \frac{1309}{5.19 - \log SiO_2} - 273.15$$

Quartz (adiabatic) $$TQA = \frac{1522}{5.75 - \log SiO_2} - 273.15$$

Chalcedony $$TCH = \frac{1032}{4.69 - \log SiO_2} - 273.15$$

Alpha-cristobalite $$TCR = \frac{1000}{4.78 - \log SiO_2} - 273.15$$

Amorphous silica $$TAM = \frac{731}{4.52 - \log SiO_2} - 273.15$$

All concentrations are in mg/l or ppm and results are given in degrees C.

Cations

NA-K $\quad TNAK = \dfrac{1217}{\log(Na/K) + 1.483} - 273.15$

NA-K-CA $\quad$ T13 or T43 $= \dfrac{1647}{\log(Na/K)+B*\log((\text{sqrt } Ca)/Na)+2.24} - 273.15$

Where B is 1/3 and 4/3 for T13 and T43 respectively.
All concentrations are in molality and results are given in degrees C.

A correction is made to the NA-K-CA geothermometers for the Mg content of the solution if the ratio of Mg/(Mg+Ca+K) (in equivalent percent) is greater than 0.5 and less than 50. The criteria, procedure and equations for application of this correction are given in Fournier and Potter (1978). The equations used for the Mg correction are as follows:

For 5. < R < 50.

DMG13 or DMG43

$$= 10.66 - 4.7415xR + 325.9(\log R)^2 - \frac{103210(\log R)^2}{TKb} - \frac{1.97x10^7(\log R)^2}{(TKb)^2} + \frac{1.61x10^7(\log R)^3}{(TKb)^2}$$

For 0.5 < R < 5.

DMG13 or DMG43

$$= -1.03 + 59.97(\log R) + 145.05(\log R)^2 - \frac{36711.(\log R)^2}{TKb} - \frac{1.67x10^7(\log R)}{(TKb)^2}$$

Where R = 100 x MGE/(MGE+CAE+KE)
MGE = Mg/molecular weight Mg
CAE = Ca/(molecular weight Ca/2)
KE = K/molecular weight K
TKb = Na-K-CA(1/3) or Na-K-Ca(4/3) temperature in Kelvin for b equal to 1/3 or 4/3 respectively

The equation used for calculation of total dissolved solids is given below:

CTDS= HCO3/2.03+SIO2+NA+K+CA+MG+CL+CO3+SO4+B+F

All concentrations are in mg/l or ppm.

EXPLANATION OF OUTPUT

Three possible outputs of the geothermometer program are shown in figures 1a, 1b, and 1c. In all cases four sections are given for each output record: site identification and location, physical properties (temperature, flow, etc.), water chemistry, and geothermometers. Temperatures given under 'Mg corrected' in figure 1a are the results for the Na-K-Ca geothermometers, if adjusted for the Mg content. When this adjustment is made, a reduction in geothermometer temperature occurs. If the Mg correction is less than zero it is not used. If the Mg content is very low relative to the other constituents (i.e. Mg/(Mg+Ca+K) < 0.5, in equivalent percent) no correction is calculated. In either case, 'none' will appears under 'Mg corrected' (Figure 1b). If the Mg content is relatively high, (i.e. Mg/(Mg+Ca+K) > 50, in equivalent percent), Fournier and Potter (1978) suggest that the measured temperature probably represents the point at which the solution equilibrated. Therefore higher values derived from Na-K-Ca geothermometers calculations are invalid. In this case the word 'cool' will appear under 'Mg corrected' (Figure 1c) and the measured temperature is likely to be the maximum temperature.

The structure of the Sample File allows numeric fields, such as the results of chemical analyses, to be qualified and the presence of a qualifier will cancel calculation of a geothermometer. Possible qualifiers are G (greater than), L (less than), R (mid-point of a range), E (estimate), N (not detected), T (trace) or Q (qualified, further explanation included in record). Qualifiers precede numeric observations and are separated from them by at least one blank (Figure 1a).

```
RECORD NO........    15370
NAME OF SOURCE.. HOT WELLS
STATE........... TEXAS
COUNTY.......... HUDSPETH

TOWNSHIP  RANGE  SECTION


WELL DEPTH(M)..    305.
FLOW RATE......
WATER TEMPERATURE...    40.0

WATER CHEMSITRY  (MG/L)
   SIO2...         20.00     CL..... L      2.00     F......          2.40
   NA.....        105.00     HCO3...      293.00     PH.....       8.8
   K......          1.60     CO3....        0.0      TOTAL DISSOLVED SOLIDS
   CA.....          4.00     SO4....       16.00     RECORDED....
   MG.....          1.80     B......        0.40     CALCULATED..        296.

GEOTHERMOMETERS  (DEGREES C)
   SILICA - CONDUCTIVE....    63.       NA-K...........     72.
            ADIABATIC.....    69.       NA-K-CA(1/3)...    101.
            CHALCEDONY....    31.       NA-K-CA(4/3)...     74.
            CRISTOBALITE..    14.       MG CORRECTED
            AMORPHOUS.....   -46.       NA-K-CA(1/3)...     43.
                                        NA-K-CA(4/3)...     53.
```

Figure 1a. Output of geothermometer program. Adjusted cation geothermometers appear under 'MG CORRECTED'. CL is an example of a qualified field. The L preceeding the 2.00 indicates that there were less than 2.00 mg/l of Cl in the solution.

RECORD NO....... 1469
NAME OF SOURCE.. UNNAMED WELL WEST OF HUECO AIRPORT (1)
STATE........... TEXAS
COUNTY.......... EL PASO

TOWNSHIP RANGE SECTION

WELL DEPTH(M).. 146.
FLOW RATE......
WATER TEMPERATURE... 38.

WATER CHEMSITRY (MG/L)

SIO2...	29.00	CL.....	33.00	F......	0.90
NA.....	97.80	HCO3...	158.00	PH.....	8.2
K......	1.30	CO3....	0.0	TOTAL DISSOLVED SOLIDS	
CA.....	2.80	SO4....	32.80	RECORDED....	285.
MG.....	0.20	B......	0.06	CALCULATED..	276.

GEOTHERMOMETERS (DEGREES C)

SILICA -	CONDUCTIVE....	76.	NA-K...........	66.
	ADIABATIC.....	82.	NA-K-CA(1/3)...	97.
	CHALCEDONY....	47.	NA-K-CA(4/3)...	74.
	CRISTOBALITE..	28.	MG CORRECTED	
	AMORPHOUS.....	-34.	NONE	

Figure 1b. The word 'NONE' appearing below 'MG CORRECTED' indicates that either the magnesium correction producded a negative value or the ratio of Mg/Mg+Ca+K was less than 0.5.

RECORD NO....... 17623
NAME OF SOURCE.. HONEY POT
STATE........... UTAH
COUNTY.......... GARFIELD

TOWNSHIP	RANGE	SECTION
36S	011E	06

WELL DEPTH(M) ..
FLOW RATE....... E 7.6
WATER TEMPERATURE... 23.

WATER CHEMSITRY (MG/L)

SIO2...	11.00	CL.....	15.00	F......	0.40
NA.....	300.00	HCO3...	405.00	PH.....	
K......	13.00	CO3....	0.0	TOTAL DISSOLVED SOLIDS	
CA.....	190.00	SO4....	1800.00	RECORDED....	2790.
MG.....	260.00	B......	0.12	CALCULATED..	2780.

GEOTHERMOMETERS (DEGREES C)

SILICA - CONDUCTIVE....	42.	NA-K...........	122.
ADABATIC.......	50.	NA-K-CA(1/3)...	131.
CHALCEDONY....	10.	NA-K-CA(4/3)...	70.
CRISTOBALITE..	-6.	MG CORRECTED	
AMORPHOUS.....	-63.	COOL	

Figure 1c. The work 'COOL' appearing below 'MG CORRECTED' indicates that the ratio of Mg/Mg+CA+K was greater than 50.

Whenever a geothermometer can not be calculated, either because a critical solute was not present in the analysis or one of the solutes is qualified, zero (0.0) will be printed in the output for that geothermometer.

PROGRAM EXPLANATION

The program consists of three loops. The first reads the input and stores it in arrays. The second loop carries out the necessary computations on all input records and the third writes the output. The program can handle up to 1000 records, however, this amount can easily be increased or decreased by changing the dimension of the arrays. If the program is to be used on substantially fewer than 1000 input records it would be advisable to decrease the dimension so that large amounts of unused core space would not be allocated.

In order to calculate geothermometers, the minimum input requirements of the program are SiO_2, Na, K, and Ca. Mg concentration is needed if that correction is used. Other input variables are not necessary for the calculations and were included to fill the needs of the low-temperature assessment work. Some or all these input variables may be eliminated from the program or not entered as input without affecting of the geothermometer calculations. However, some factors such as pH are very useful for evaluating the validity of the results. For example, the silica geothermometers are not valid for pH greater than 8.5.

DEFINITION OF VARIABLES

Input Variables

NUM - GEOTHERM record number. 'I' format, maximum length = 7

NAME1, NAME2, NAME3, NAME4, NAME5 - Name of sample source, that is the name of the spring or well from which the sample was taken. The NAME variables are read using an 'A' format, therefore any alphanumeric character is an acceptable value for the variables and, taken together, can handle a name up to 40 characters in length.

'A' format, maximum length = 8 (total of 40)

STATE1, STATE2 - The name of the state in which the sample site is located. 'A' format, maximum length= 8 (total of 16)

COUN1, COUN2 - The name of the county in which the sample site is located. 'A' format, maximum length = 8 (total of 16)

TWNS - Township. 'A' format, maximum length = 3

RANGE - Range. 'A' format, maximum length = 4

SECT - Section. 'A' format, maximum length = 3

WELLD - The depth of well if sample is from a well. 'A' format, maximum length = 7

FLOW, RATE - FLOW is the discharge of the sample and RATE is the units used in measuring it. FLOW, 'A' format, maximum length = 8
RATE, 'A' format, maximum length = 8

WTEMP - Measured temperature of the sample. All GEOTHERM temperatures are given in degrees C. 'A' format, maximum length = 6

UNITS - Units used for measuring solutes in sample. 'A' format, maximum length = 4

PH - pH of sample. 'A' format, maximum length = 6

SI02, NA, K, CA, MG, F, CL, HC03, C03, S04, B - Solutes measured in sample, each variable is the same as the chemical abbreviation (Except all letters are upper case). All are real variables and must include a decimal point.
'F' format, maximum length = 8.2 (Two places to the right of the decimal)

QSI02, QNA, QK, QCA, QMG, QF, QCL, QHC03, QC03, QS04, QB - Qualifier for each solute. These variables are necessary because no qualified value is used in the geothermometer calculations, qualified fields are treated by the program as if they have a value of zero. 'A' format, maximum length = 1

RTDS - Recorded total dissolved solids (i.e. any value listed in the GEOTHERM record for dissolved solids). 'A' format, maximum length = 8

Calulated Variables

CTDS - Total dissolved solids as calculated by this program from concentrations of the chemical variables entered as input. The resultant value of the calculation is rounded to the nearest ten if it exceeds 1000. 'F' format, maximum length = 8.0

TQC - Silica geothermometer, quartz conductive. 'F' format, maximum length = 6.0

TCH - Silica geothermometer, chalcedony. 'F' format, maximum length = 6.0

TQA - Silica geothermometer, quartz adiabatic. 'F' format, maximum length = 6.0

TCR - Silica geothermometer, alpha-cristobalite. 'F' format, maximum length = 6.0

TAM - Silica geothermometer, amorphous silica. 'F' format, maximum length = 6.0

TNAK - NA-K geothermometer. 'F' format, maximum length = 6.0

T13 - NA-K-CA (1/3) geothermometer, uncorrected. 'F' format, maximum length = 6.0

T43 - NA-K-CA (4/3) geothermometer, uncorrected. 'F' format, maximum length = 6.0

TMG13 - NA-K-CA (1/3) geothermometer, corrected for MG. 'F' format, maximum length = 6.0

TMG43 - NA-K-CA (4/3) geothermometer, corrected for MG. 'F' format, maximum length = 6.0

Miscellaneous Variable

LABEL - This variable is used as a flag which keeps track of status of the MG corrections for each record. The value of label determines the format to be used in the 'MG corrected' portion of the output. Integer variable, possible values = 0,1,2,3,4

INPUT FORMAT

Consists of two lines, each having a blank in column one.

Line one = NUM cols 2-8
NAME1, NAME2, NAME3, NAME4, NAME5 cols 9-48
STATE1, STATE2 cols 49-64
COUN1,COUN2 cols 65-80
TWNS cols 81-83
RANGE cols 84-87
SECT cols 88-90
WELLD cols 91-97
FLOW cols 98-115
RATE cols 116-121
WTEMP cols 122-127
UNITS cols 128-131
PH cols 132-137

Line two =	QSIO2	col 2
	SIO2	cols 4-11
	QNA	col 12
	NA	cols 14-21
	QK	col 22
	K	cols 24-31
	QCA	col 32
	CA	cols 34-41
	QMG	col 42
	MG	cols 44-51
	QCL	col 52
	CL	cols 54-61
	QHCO3	col 62
	HCO3	cols 64-71
	QCO3	col 72
	CO3	cols 74-81
	QSO4	col 82
	SO4	cols 84-91
	QB	col 92
	B	cols 94-101
	QF	col 102
	F	cols 104-111
	RTDS	cols 112-119

PROGRAM LISTINGS

A listing of the original program, which uses Sample File input is given in Appendix A. Appendix B lists a modified version of the program which can be used on more generalized input data. The portions of the original program that were altered or removed to create the revised version are outlined in Appendix A. The major effect of the revisions is to reduce the number of input and output variables and to eliminate all lines that deal with the Sample File qualifiers. In the new version of the program, the input and output consists of the concentration of the solutes which are required to calculate the geothermometers and some basic identifying information (eg. Sample number, source name, state, county, temperature and pH). Appendix B also contains a sample of an input file for the revised version of the program and output produced by it.

The program listings follow. Any line with a 'C' in column one is a comment and a '1' in column six indicates the line is a continuation of the preceeding line.

Appendix C is an annotated listing of the Job Control Language (JCL) and Gipsy commands which must surround the program if it is run on the Amdahl computer system in Reston and GEOTHERM is used as the source of input data.

REFERENCES CITED

Ellis, A. J., and Mahon, W. A. J., 1977, Chemistry and geothermal systems: Academic Press, New York, 392 p.

Fournier, R. O., 1977, Chemical geothermometers and mixing models for geothermometers and mixing models for geothermal systems: Geothermics, v. 5, p. 41-50.

Fournier, R. O., 1981, Application of water geochemistry to geothermal exploration and reservoir engineering; in Geothermal systems - principles and case histories: John Wiley and Sons, New York, p. 109-143.

Fournier, R. O. and Potter II, R. W., 1978, A magnesium correction for the Na-Ca-K geothermometers: Geochimica et Cosmochimica Acta, v. 43, 1543-1550.

APPENDIX C

```
   (job name)  (acct. num)          (user name)    (class in descending proirity
                                                            may be H,A,B,C,D,E,F)
//MG9279MN JOB ($$$$$$$$$,B200,15,100),'RAPPORT',CLASS=A
//A EXEC QUESTRAN,DNAME='RIF.W0040.GEOD1',DVOL=CCD812,
//         DUNIT='3330-1',RNAME='RIF.W0020.THERM1',RVOL=CCD821,
//         UNIT='3330-1', CLOCK=15,SPACE=800,RGN=110K,LINE=132
//QUESTRAN.SYSWRKO DD DSN=&&TEMP,UNIT=SYSDK,DISP=(NEW,PASS),
//         DCB=(RECFM=FB,LRECL=132,BLKSIZE=132),
//        SPACE=(CYL,(10,1),RLSE)
//QUESTRAN.SYSRDR DD *
FORM
 SAMPLE
SELECT
 A. B60<UTAH>   (Gipsy select variable and logic statement)
 LOGIC A
SORT
 B60 20
 B65 20
 B20 25
COPY
 ' '
 A10 7
 B20 40
 B60 16
 B65 16
 B95 3
 B105 4
 B115 3
 M25 7
 M220 8
 M223 6
 M210 6
 M341 4
 M20 6
 ' '
 NEW RECORD
 ' '
 M130 10
 M40 10
 M50 10
 M60 10
 M70 10
 M80 10
 M140 10
 M580 10
 M110 10
 M120 10
 M90 10
 M23 8
/*
//B EXEC FTG1CLG,REGION.FORT=120K,REGION.GO=450K     (FORTRAN compiler, load and go.)
//FORT.SYSIN DD *
     program

/*
//GO.FT02F001 DD DSN=&&TEMP,UNIT=SYSDK,DISP=(OLD,DELETE)         (Input to program)
//GO.FT04F001 DD SYSOUT=A,DCB=(RECFM=FAB,LRECL=133,BLKSIZE=12901)  (Output from program)
/*
```

Original Geothermometer program.

```
      IMPLICIT REAL*8 (A-Z)
      DIMENSION NAME1(1000),NAME2(1000),NAME3(1000),NAME4(1000)
      DIMENSION STATE1(1000),STATE2(1000),COUN1(1000),COUN2(1000)
      DIMENSION TWNS(1000),SECT(1000),WELLD(1000),RANGE(1000)
      DIMENSION FLOW(1000),RATE(1000),WTEMP(1000),UNITS(1000),PH(1000)
      DIMENSION SIO2(1000),NA(1000),K(1000),CA(1000),MG(1000),F(1000)
      DIMENSION CL(1000),HCO3(1000),CO3(1000),SO4(1000),B(1000)
      INTEGER*2 QSIO2(1000),QNA(1000),QK(1000),QCA(1000),QMG(1000)
      INTEGER*2 QCL(1000),QHCO3(1000),QCO3(1000),QSO4(1000),QB(1000)
      INTEGER*2 QF(1000),BLANK
      DIMENSION RTDS(1000),CTDS(1000),TQC(1000),TCH(1000)
      DIMENSION TQA(1000),TCR(1000),NAME5(1000)
      DIMENSION TAM(1000),TNAK(1000),T13(1000),T43(1000),TMG13(1000)
     1,TMG43(1000)
      INTEGER I,LABEL(1000),NUM(1000),J,M,N,L
      DATA BLANK/1H]/
      TK=273.15
C-----------------------------------------------------------------------
C-- MWNA=MOLECULAR WEIGHT NA, MWK=MOLECULAR WEIGHT K, MWCA=MOLECULAR ---
C--          WEIGHT CA AND MWMG=MOLECULAR WEIGHT MG                   ---
C-----------------------------------------------------------------------
      MWNA=23.
      MWK=39.1
      MWCA=40.08
      MWMG=12.16
C-----------------------------------------------------------------------
C--               READ VALUES OF ALL VARIABLES                        ---
C-----------------------------------------------------------------------
      DO 5 J=1,1000
      READ(2,10,END=30) NUM(J),NAME1(J),NAME2(J),NAME3(J),NAME4(J),
     1NAME5(J),
     1STATE1(J),STATE2(J),COUN1(J),COUN2(J),TWNS(J),RANGE(J),SECT(J),
     1WELLD(J),FLOW(J),RATE(J),WTEMP(J),UNITS(J),PH(J)
   10 FORMAT(1X,I7,9A8,A3,A4,A3,A7,A8,A6,A6,A4,A6)
      IF (NUM(J).EQ.0) GO TO 30
C-----------------------------------------------------------------------
C-- INITIALIZE ALL QUALIFIERS TO BLANK WHICH CONSISTS OF A NONPRINTING--
C--                          PAD CHARACTER                            ---
C-----------------------------------------------------------------------
      QSIO2(J)=BLANK
      QNA(J)=BLANK
      QK(J)=BLANK
      QMG(J)=BLANK
      QCL(J)=BLANK
      QHCO3(J)=BLANK
      QSO4(J)=BLANK
      QB(J)=BLANK
      QF(J)=BLANK
      QCA(J)=BLANK
      QCO3(J)=BLANK
C-----------------------------------------------------------------------

C--       READ VALUES OF ALL SPECIES AND THEIR QUALIFIERS             ---
C-----------------------------------------------------------------------
```

```
      READ(2,25) QSIO2(J),SIO2(J),QNA(J),NA(J),QK(J),K(J),QCA(J),CA(J),
     1QMG(J),MG(J),QCL(J),CL(J),QHCO3(J),HCO3(J),QCO3(J),CO3(J),QSO4(J),
     1SO4(J),QB(J),B(J),QF(J),F(J),RTDS(J)
   25 FORMAT(1X,11(A1,1X,F8.2),A8)
      L = J
    5 CONTINUE
   30 DO 80 M=1,L
      LABEL(M)=0
      XHCO3=HCO3(M)
      XSIO2=SIO2(M)
      XNA=NA(M)
      XK=K(M)
      XCA=CA(M)
      XMG=MG(M)
      XCL=CL(M)
      XCO3=CO3(M)
      XSO4=SO4(M)
      XB=B(M)
      XF=F(M)
C---------------------------------------------------------------------
C--     CHECK FOR QUALIFIERS AND SET INTERMEDIATE VALUES TO ZERO    ---
C--     QUALIFIED VALUES ARE NOT USED IN ANY OF THE GEOTHERMOMETER  ---
C--                          CALCULATIONS                           ---
C---------------------------------------------------------------------
      IF(QHCO3(M).NE.BLANK) XHCO3=0.
      IF(QSIO2(M).NE.BLANK) XSIO2=0.
      IF(QNA(M).NE.BLANK) XNA=0.
      IF(QK(M).NE.BLANK) XK=0.
      IF(QCA(M).NE.BLANK) XCA=0.
      IF(QMG(M).NE.BLANK) XMG=0.
      IF(QCL(M).NE.BLANK) XCL=0.
      IF(QCO3(M).NE.BLANK) XCO3=0.
      IF(QSO4(M).NE.BLANK) XSO4=0.
      IF(QB(M).NE.BLANK) XB=0.
      IF(QF(M).NE.BLANK) XF=0.
C---------------------------------------------------------------------
C--                 CALCULATE TOTAL DISSOLVED SOLIDS                ---
C--  VALUES OF TOTAL DISSOLVED SOLIDS GREATER THAN 1000 ARE ROUNDED  ---
C--                  TO THE NEAREST 10 PPM OR MG/L                  ---
C---------------------------------------------------------------------
      CTDS(M) = (XHCO3/2.03)+XSIO2+XNA+XK+XCA+XMG+XCL+XCO3+XSO4+XB+XF
      IF(CTDS(M).LT.1000.) GO TO 50
      I = CTDS(M)/10
      CTDS(M) = I*10.
   50 IF(SIO2(M).LE.0..OR.XSIO2.LE.0.) GO TO 55
C---------------------------------------------------------------------
C--                 CALCULATE SILICA GEOTHERMOMETERS                ---
C---------------------------------------------------------------------
      LSIO2=DLOG10(SIO2(M))
      TQC(M)=-TK+1309./(5.19-LSIO2)
      TQA(M)=-TK+1522./(5.75-LSIO2)
      TCH(M)=-TK+1032./(4.69-LSIO2)
      TCR(M)=-TK+1000./(4.78-LSIO2)
      TAM(M)=-TK+731./(4.52-LSIO2)
```

```
      GO TO 91
   55 TQC(M)=0.
      TQA(M)=0.
      TCH(M)=0.
      TAM(M)=0.
      TCR(M)=0.
   91 IF(NA(M).LE.0..OR.K(M).LE.0.) GO TO 95
      IF(XNA.LE.0..OR.XK.LE.0.) GO TO 95
C-----------------------------------------------------------------------
C--                 CALCULATE NA-K GEOTHERMOMETERS                    ---
C-----------------------------------------------------------------------
      NAE=NA(M)/MWNA
      NAM=NAE/1000.
      KE=K(M)/MWK
      KM=KE/1000.
      LNAK=DLOG10(NAM/KM)
      TNAK(M)=-TK+1217./(LNAK+1.483)
   94 IF(CA(M).LE.0..OR.XCA.LE.0.) GO TO 82
C-----------------------------------------------------------------------
C--           CALCULATE NA-K-CA (1/3 & 4/3) GEOTHERMOMETERS           ---
C-----------------------------------------------------------------------
      CAE=CA(M)/(MWCA/2.)
      CAM=(CA(M)/MWCA)/1000.
      SCAM=DSQRT(CAM)
      LCANA=DLOG10(SCAM/NAM)
      T13K=1647./(LNAK+.333*LCANA+2.24)
      T43K=1647./(LNAK+1.333*LCANA+2.24)
      T13(M)=T13K-TK
      T43(M)=T43K-TK
      T13KS=T13K**2
      T43KS=T43K**2
C-----------------------------------------------------------------------
C--                 CALCULATE MG CORRECTIONS (1/3 & 4/3)              ---
C-----------------------------------------------------------------------
   96 IF(MG(M).LE.0..OR.XMG.LE.0.) GO TO 800
      MGE=MG(M)/MWMG
      R=100*MGE/(MGE+CAE+KE)
      IF (R.GT.50.) GO TO 98
      LR=DLOG10(R)
      LRS=LR**2
      LRC=LR**3
      IF(R.LT.0.5) GO TO 800
      IF(R.LT.5.) GO TO 92
      DMG13=10.66-4.7415*R+325.9*LRS-103210*LRS/T13K-1.97E7*LRS/T13KS
     1+1.61E7*LRC/T13KS
      DMG43=10.66-4.7415*R+325.9*LRS-103210*LRS/T43K-1.97E7*LRS/T43KS
     1+1.61E7*LRC/T43KS
      GO TO 93
   92 DMG13=-1.03+59.97*LR+145.05*LRS-36711.*LRS/T13K-1.675E7*LR/T13KS
      DMG43=-1.03+59.97*LR+145.05*LRS-36711.*LRS/T43K-1.675E7*LR/T43KS
   93 IF(DMG13.LE.0.) LABEL(M)=3
      IF(DMG43.LE.0.) LABEL(M)=4
      IF(DMG13.LE.0..and.DMG43.LE.0.) LABEL(M)=1
```

```
C-----------------------------------------------------------------------
C--     RECALCULATE NA-K-CA GEOTHERMOMETERS WITH MG CORRECTION       ---
C-----------------------------------------------------------------------
      TMG13(M)=T13(M)-DMG13
      TMG43(M)=T43(M)-DMG43
      GO TO 80
C-----------------------------------------------------------------------
C--       LABEL IS A FLAG FOR PRINTING MG CORRECTED VALUES           ---
C-----------------------------------------------------------------------
  800 LABEL(M)=1
      GO TO 80
   98 LABEL(M)=2
      GO TO 80
   95 TNAK(M)=0.
   82 T13(M)=0.
      T43(M)=0.
      LABEL(M)=1
   80 CONTINUE
C-----------------------------------------------------------------------
C--                        PRINT RESULTS                             ---
C-----------------------------------------------------------------------
      DO 500 N=1,L
   99 WRITE(4,100) NUM(N)
  100 FORMAT('1',3X,'RECORD NO....... ',I7)
      WRITE(4,105) NAME1(N),NAME2(N),NAME3(N),NAME4(N),NAME5(N)
  105 FORMAT(4X,'NAME OF SOURCE.. ',5A8)
      WRITE(4,110) STATE1(N),STATE2(N)
  110 FORMAT(4X,'STATE........... ',2A8)
      WRITE(4,120) COUN1(N),COUN2(N)
  120 FORMAT(4X,'COUNTY.......... ',2A8)
      WRITE(4,130)
  130 FORMAT('0',4X,'TOWNSHIP  RANGE  SECTION')
      WRITE(4,140) TWNS(N),RANGE(N),SECT(N)
  140 FORMAT(6X,A3,6X,A4,4X,A3)
      WRITE(4,150) WELLD(N)
  150 FORMAT('0',3X,'WELL DEPTH(M).. ',A7)
      WRITE(4,160) FLOW(N),RATE(N)
  160 FORMAT(4X,'FLOW RATE...... ',A8,3X,A6)
      WRITE(4,170) WTEMP(N)
  170 FORMAT(4X,'WATER TEMPERATURE... ',A6)
      WRITE(4,180) UNITS(N)
  180 FORMAT('0',3X,'WATER CHEMSITRY (',A4,')')
      WRITE(4,190) QSIO2(N),SIO2(N),QCL(N),CL(N),QF(N),F(N)
  190 FORMAT(7X,'SIO2... ',A1,1X,F8.2,4X,'CL..... ',A1,1X,F8.2,4X,
     1'F...... ',A1,1X,F8.2)
      WRITE(4,200) QNA(N),NA(N),QHCO3(N),HCO3(N),PH(N)
  200 FORMAT(7X,'NA..... ',A1,1X,F8.2,4X,'HCO3... ',A1,1X,F8.2,
     14X,'PH..... ',2X,A8)
      WRITE(4,205) QK(N),K(N),QCO3(N),CO3(N)
  205 FORMAT(7X,'K...... ',A1,1X,F8.2,4X,'CO3.... ',A1,1X,F8.2,
     14X,'TOTAL DISSOLVED SOLIDS')
      WRITE(4,210) QCA(N),CA(N),QSO4(N),SO4(N),RTDS(N)
  210 FORMAT(7X,'CA..... ',A1,1X,F8.2,4X,'SO4.... ',A1,1X,F8.2,
     14X,'RECORDED.... ',A8)
```

```
      WRITE(4,215) QMG(N),MG(N),QB(N),B(N),CTDS(N)
  215 FORMAT(7X,'MG..... ',A1,1X,F8.2,4X,'B...... ',A1,1X,F8.2,
     14X,'CALCULATED.. ',2X,F8.0)
      WRITE(4,220)
  220 FORMAT('0',3X,'GEOTHERMOMETERS (DEGREES C)')
      WRITE(4,230) TQC(N),TNAK(N)
  230 FORMAT(7X,'SILICA - CONDUCTIVE.... ',F6.0,8X,'NA-K........... '
     1,F6.0)
      WRITE(4,235) TQA(N),T13(N)
  235 FORMAT(16X,'ADABATIC...... ',F6.0,8X,'NA-K-CA(1/3)... ',F6.0)
      WRITE(4,240) TCH(N),T43(N)
  240 FORMAT(16X,'CHALCEDONY.... ',F6.0,8X,'NA-K-CA(4/3)... ',F6.0)
      WRITE(4,245) TCR(N)
  245 FORMAT(16X,'CRISTOBALITE.. ',F6.0,8X,'MG CORRECTED')
      IF (LABEL(N).NE.0) GO TO 400
      WRITE(4,250) TAM(N),TMG13(N)
  250 FORMAT(16X,'AMORPHOUS..... ',F6.0,8X,'NA-K-CA(1/3)... ',F6.0)
      WRITE(4,255) TMG43(N)
  255 FORMAT(45X,'NA-K-CA(4/3)... ',F6.0)
      go to 500
  400 IF(LABEL(N).EQ.1.OR.LABEL(N).EQ.2) GO TO 411
      IF(LABEL(N).EQ.3) GO TO 413
      IF(LABEL(N).EQ.4) WRITE(4,250) TAM(N),TMG13(N)
      GO TO 500
  413 WRITE(4,267) TAM(N),TMG43(N)
  267 FORMAT(16x,'AMORPHOUS..... ',F6.0,8X,'NA-K-CA(4/3)... ',F6.0)
      GO TO 500
  411 IF (LABEL(N).EQ.1) WRITE(4,260) TAM(N)
  260 FORMAT(16X,'AMORPHOUS..... ',F6.0,8X,'NONE')
      IF (LABEL(N).EQ.2) WRITE(4,265) TAM(N)
  265 FORMAT(16X,'AMORPHOUS..... ',F6.0,8X,'COOL')
  500 CONTINUE
      STOP
      END
§§§
```

APPENDIX B

Revised geothermometer program

```
      IMPLICIT REAL*8 (A-Z)
      DIMENSION NAME1(1000),NAME2(1000),NAME3(1000),NAME4(1000)
      DIMENSION STATE1(1000),STATE2(1000),COUN1(1000),COUN2(1000)
      DIMENSION WTEMP(1000),PH(1000)
      DIMENSION SIO2(1000),NA(1000),K(1000),CA(1000),MG(1000)
      DIMENSION TQC(1000),TCH(1000)
      DIMENSION TQA(1000),TCR(1000),NAME5(1000)
      DIMENSION TAM(1000),TNAK(1000),T13(1000),T43(1000),TMG13(1000)
     1,TMG43(1000)
      INTEGER I,LABEL(1000),NUM(1000),J,M,N,L
      TK=273.15
C------------------------------------------------------------------------
C-- MWNA=MOLECULAR WEIGHT NA, MWK=MOLECULAR WEIGHT K, MWCA=MOLECULAR ---
C--          WEIGHT CA AND MWMG=MOLECULAR WEIGHT MG                  ---
C------------------------------------------------------------------------
      MWNA=23.
      MWK=39.1
      MWCA=40.08
      MWMG=12.16
C------------------------------------------------------------------------
C--             READ VALUES OF ALL VARIABLES                         ---
C------------------------------------------------------------------------
      DO 5 J=1,1000
      READ(2,10,END=30) NUM(J),NAME1(J),NAME2(J),NAME3(J),NAME4(J),
     1NAME5(J),
     1STATE1(J),STATE2(J),COUN1(J),COUN2(J)
     1,WTEMP(J),PH(J)
   10 FORMAT(1X,I7,9A8,A6,A6)
      IF (NUM(J).EQ.0) GO TO 30
C------------------------------------------------------------------------
C--             READ VALUES OF ALL SPECIES                           ---
C------------------------------------------------------------------------
      READ(2,25) SIO2(J),NA(J),K(J),CA(J),MG(J)
   25 FORMAT(1X,5(2X,F8.2))
      L = J
    5 CONTINUE
C------------------------------------------------------------------------
C--                    CALCULATION LOOP                              ---
C------------------------------------------------------------------------
   30 DO 80 M=1,L
      LABEL(M)=0
   50 IF(SIO2(M).LE.0.) GO TO 55
C------------------------------------------------------------------------
C--               CALCULATE SILICA GEOTHERMOMETERS                   ---
C------------------------------------------------------------------------
      LSIO2=DLOG10(SIO2(M))
      TQC(M)=-TK+1309./(5.19-LSIO2)
      TQA(M)=-TK+1522./(5.75-LSIO2)
      TCH(M)=-TK+1032./(4.69-LSIO2)
      TCR(M)=-TK+1000./(4.78-LSIO2)
      TAM(M)=-TK+731./(4.52-LSIO2)
```

```
      GO TO 91
   55 TQC(M)=0.
      TQA(M)=0.
      TCH(M)=0.
      TAM(M)=0.
      TCR(M)=0.
   91 IF(NA(M).LE.0..OR.K(M).LE.0.) GO TO 95
C-----------------------------------------------------------------------
C--              CALCULATE NA-K GEOTHERMOMETERS                     ---
C-----------------------------------------------------------------------
      NAE=NA(M)/MWNA
      NAM=NAE/1000.
      KE=K(M)/MWK
      KM=KE/1000.
      LNAK=DLOG10(NAM/KM)
      TNAK(M)=-TK+1217./(LNAK+1.483)
   94 IF(CA(M).LE.0.) GO TO 82
C-----------------------------------------------------------------------
C--        CALCULATE NA-K-CA (1/3 & 4/3) GEOTHERMOMETERS            ---
C-----------------------------------------------------------------------
      CAE=CA(M)/(MWCA/2.)
      CAM=(CA(M)/MWCA)/1000.
      SCAM=DSQRT(CAM)
      LCANA=DLOG10(SCAM/NAM)
      T13K=1647./(LNAK+.333*LCANA+2.24)
      T43K=1647./(LNAK+1.333*LCANA+2.24)
      T13(M)=T13K-TK
      T43(M)=T43K-TK
      T13KS=T13K**2
      T43KS=T43K**2
C-----------------------------------------------------------------------
C--              CALCULATE MG CORRECTIONS (1/3 & 4/3)               ---
C-----------------------------------------------------------------------
   96 IF(MG(M).LE.0.) GO TO 800
      MGE=MG(M)/MWMG
      R=100*MGE/(MGE+CAE+KE)
      IF (R.GT.50.) GO TO 98
      LR=DLOG10(R)
      LRS=LR**2
      LRC=LR**3
      IF(R.LT.0.5) GO TO 800
      IF(R.LT.5.) GO TO 92
      DMG13=10.66-4.7415*R+325.9*LRS-103210*LRS/T13K-1.97E7*LRS/T13KS
     1+1.61E7*LRC/T13KS
      DMG43=10.66-4.7415*R+325.9*LRS-103210*LRS/T43K-1.97E7*LRS/T43KS
     1+1.61E7*LRC/T43KS
      GO TO 93
   92 DMG13=-1.03+59.97*LR+145.05*LRS-36711.*LRS/T13K-1.675E7*LR/T13KS
      DMG43=-1.03+59.97*LR+145.05*LRS-36711.*LRS/T43K-1.675E7*LR/T43KS
   93 IF(DMG13.LE.0.) LABEL(M)=3
      IF(DMG43.LE.0.) LABEL(M)=4
      IF(DMG13.LE.0..and.DMG43.LE.0.) LABEL(M)=1
C-----------------------------------------------------------------------
C--     RECALCULATE NA-K-CA GEOTHERMOMETERS WITH MG CORRECTION      ---
```

```
C-----------------------------------------------------------------------
      TMG13(M)=T13(M)-DMG13
      TMG43(M)=T43(M)-DMG43
      GO TO 80
C-----------------------------------------------------------------------
C--       LABEL IS A FLAG FOR PRINTING MG CORRECTED VALUES           ---
C-----------------------------------------------------------------------
  800 LABEL(M)=1
      GO TO 80
   98 LABEL(M)=2
      GO TO 80
   95 TNAK(M)=0.
   82 T13(M)=0.
      T43(M)=0.
      LABEL(M)=1
   80 CONTINUE
C-----------------------------------------------------------------------
C--                          PRINT RESULTS                           ---
C-----------------------------------------------------------------------
      DO 500 N=1,L
      WRITE(4,501)
  501 FORMAT('0','-----------------------------------------------------
     1------------------')
   99 WRITE(4,100) NUM(N)
  100 FORMAT('0',3X,'RECORD NO....... ',I7)
      WRITE(4,105) NAME1(N),NAME2(N),NAME3(N),NAME4(N),NAME5(N)
  105 FORMAT(4X,'NAME OF SOURCE.. ',5A8)
      WRITE(4,110) STATE1(N),STATE2(N)
  110 FORMAT(4X,'STATE........... ',2A8)
      WRITE(4,120) COUN1(N),COUN2(N)
  120 FORMAT(4X,'COUNTY.......... ',2A8)
      WRITE(4,170) WTEMP(N),PH(N)
  170 FORMAT('0',3X,'WATER TEMPERATURE... ',A6,4X,'PH..... ',A8)
      WRITE(4,502)
  502 FORMAT('0',3X,'WATER CHEMISTRY')
      WRITE(4,190) SIO2(N)
  190 FORMAT(7X,'SIO2... ',2X,F8.2)
      WRITE(4,200) NA(N)
  200 FORMAT(7X,'NA..... ',2X,F8.2)
      WRITE(4,205) K(N)
  205 FORMAT(7X,'K...... ',2X,F8.2)
      WRITE(4,210) CA(N)
  210 FORMAT(7X,'CA..... ',2X,F8.2)
      WRITE(4,215) MG(N)
  215 FORMAT(7X,'MG..... ',2X,F8.2)
      WRITE(4,220)
  220 FORMAT('0',3X,'GEOTHERMOMETERS (DEGREES C)')
      WRITE(4,230) TQC(N),TNAK(N)
  230 FORMAT(7X,'SILICA - CONDUCTIVE.... ',F6.0,8X,'NA-K........... '
     1,F6.0)
      WRITE(4,235) TQA(N),T13(N)
  235 FORMAT(16X,'ADABATIC...... ',F6.0,8X,'NA-K-CA(1/3)... ',F6.0)
      WRITE(4,240) TCH(N),T43(N)
  240 FORMAT(16X,'CHALCEDONY.... ',F6.0,8X,'NA-K-CA(4/3)... ',F6.0)
```

```
      WRITE(4,245) TCR(N)
  245 FORMAT(16X,'CRISTOBALITE.. ',F6.0,8X,'MG CORRECTED')
      IF (LABEL(N).NE.0) GO TO 400
      WRITE(4,250) TAM(N),TMG13(N)
  250 FORMAT(16X,'AMORPHOUS..... ',F6.0,8X,'NA-K-CA(1/3)... ',F6.0)
      WRITE(4,255) TMG43(N)
  255 FORMAT(45X,'NA-K-CA(4/3)... ',F6.0)
      GO TO 500
  400 IF(LABEL(N).EQ.1.OR.LABEL(N).EQ.2) GO TO 411
      IF(LABEL(N).EQ.3) GO TO 413
      IF(LABEL(N).EQ.4) WRITE(4,250) TAM(N),TMG13(N)
      GO TO 500
  413 WRITE(4,267) TAM(N),TMG43(N)
  267 FORMAT(16x,'AMORPHOUS..... ',F6.0,8X,'NA-K-CA(4/3)... ',F6.0)
      GO TO 500
  411 IF (LABEL(N).EQ.1) WRITE(4,260) TAM(N)
  260 FORMAT(16X,'AMORPHOUS..... ',F6.0,8X,'NONE')
      IF (LABEL(N).EQ.2) WRITE(4,265) TAM(N)
  265 FORMAT(16X,'AMORPHOUS..... ',F6.0,8X,'COOL')
  500 CONTINUE
      STOP
      END
```

Sample Input for Revised Version of Program

```
17623  HONEY POT                              UTAH          GARFIELD       23.
11.       300.      13.       190.      260.
17311  ELBERTA                                UTAH          UTAH           22.    6.8
70.       52.       15.       278.      101.
 1469  UNAMED WELL WEST OF HUECO AIRPORT (1)  TEXAS         EL PASO        38.    8.2
29.       97.8      1.3       2.8       0.2
```

```
------------------------------------------------------------------------------

   RECORD NO.......    17311
   NAME OF SOURCE.. ELBERTA
   STATE........... UTAH
   COUNTY.......... UTAH

   WATER TEMPERATURE...   22.     PH.....    6.8

   WATER CHEMISTRY
      SIO2...        70.00
      NA.....        52.00
      K......        15.00
      CA.....       278.00
      MG.....       101.00

   GEOTHERMOMETERS (DEGREES C)
      SILICA - CONDUCTIVE....    118.       NA-K...........    267.
               ADABATIC......    117.       NA-K-CA(1/3)...    193.
               CHALCEDONY....     90.       NA-K-CA(4/3)...     50.
               CRISTOBALITE..     68.       MG CORRECTED
               AMORPHOUS.....      0.       NA-K-CA(1/3)...     38.

------------------------------------------------------------------------------

   RECORD NO.......    17623
   NAME OF SOURCE.. HONEY POT
   STATE........... UTAH
   COUNTY.......... GARFIELD

   WATER TEMPERATURE...   23.     PH.....

   WATER CHEMISTRY
      SIO2...        11.00
      NA.....       300.00
      K......        13.00
      CA.....       190.00
      MG.....       260.00

   GEOTHERMOMETERS (DEGREES C)
      SILICA - CONDUCTIVE....     42.       NA-K...........    122.
               ADABATIC......     50.       NA-K-CA(1/3)...    131.
               CHALCEDONY....     10.       NA-K-CA(4/3)...     70.
               CRISTOBALITE..     -6.       MG CORRECTED
               AMORPHOUS.....    -63.       COOL

------------------------------------------------------------------------------

   RECORD NO.......     1469
   NAME OF SOURCE.. UNNAMED WELL WEST OF HUECO AIRPORT (1)
   STATE........... TEXAS
   COUNTY.......... EL PASO

   WATER TEMPERATURE...   38.     PH.....    8.2

   WATER CHEMISTRY
      SIO2...        29.00
      NA.....        97.80
      K......         1.30
```

CA.....	2.80
MG.....	0.20

GEOTHERMOMETERS (DEGRFES C)

SILICA -	CONDUCTIVE....	78.	NA-K...........	66.
	ADABATIC......	82.	NA-K-CA(1/3)...	97.
	CHALCEDONY....	47.	NA-K-CA(4/3)...	74.
	CRISTOBALITE..	26.	MG CORRECTED	
	AMORPHOUS.....	-34.	NONF	